ReportViewer Demo – IIS Log Analyzer

Step 1 – Add log parser class
1.1
Create a new VB Windows Application project

1.2
From the menu choose Project -> Add Reference. In the Add Reference dialog choose System.Web and click OK.

1.3
From the menu choose Project -> Add Class.

1.4
In the Add New Item dialog, enter LogParser.vb as the file name.

1.5
Paste code from Appendix A.

Step 2 – Create data source
2.1
From Data menu choose ‘Add New Data Source’.

2.2
In Data Source Configuration Wizard choose Object, and then click ‘Next’.
2.3
In ‘Select the Object you wish to bind to’ screen expand the tree and choose ‘LogFileEntry’ and click Finish.
Step 3 – Design the report

3.1
From the menu choose Project > Add New Item.
3.2
In Add New Item dialog choose Report and click Add. This launches Report Designer.
3.3
If Data Sources window is not visible then choose Show Data Sources from Data menu.

3.4
From the toolbox drag and drop a table to the report.
3.5
From Data Sources window drag and drop the Document field into the detail row of the first column in the table. The detail row is the middle row. Notice the header row is automatically filled.

3.6
Right-click on the row selector of the detail row and choose ‘Edit Group’. In the Details Grouping dialog, in the ‘Group on’ list, click on the first row, then from the dropdown choose =Fields!Document.Value then click OK to close the dialog.
3.7
In the Detail row of the second column enter =CountRows() and in the header of this column enter the column name: Hits.

3.8
Delete the last column of the table by right-clicking on the column selector and choosing Delete Columns. Increase the width of the first column by dragging its right edge in the column selector.

Step 4 – Add the report to the form

4.1
Switch to the Form, then from the Toolbox drag & drop ReportViewer control to the Form. You can find ReportViewer in the Data section.
4.2
Open the Smart Tags Panel of the ReportViewer control. Click the ‘Choose Report’ dropdown and select the report you designed. Notice that an instance of LogFileEntryBindingSource appears in the component tray. Optionally, dock the ReportViewer control to the form.
4.3
From the menu choose View > Code. As the first line of the Form1_Load method, enter this line:

Me.LogFileEntryBindingSource.DataSource = _

 LogParser.ParseLogFile("c:\\sample.log")
Step 5 – Run the application

5.1
Press F5 to run the application. 
The report shows a list of documents accessed, and the number of times each document was accessed. This list is not sorted, and all documents are listed, including ones that have only been accessed once. In the next step we will modify the report to only show the most popular documents, and show them in sorted order.
Step 6 – Modify report to show top 10 most popular documents
6.1
Right-click on the row selector of the Detail row and choose ‘Edit Group’. In the Details Grouping dialog switch to Filters tab. In the Expression column enter =CountRows() and in Operator column select Top N. In the Value column type =10 and then click OK to close the dialog.
Run application and notice that only the top 10 most popular documents are displayed. However they are not listed in sorted order. We’ll fix that in the next step.
6.2
Right-click on the table selector and choose ‘Properties’. In the Table Properties dialog switch to Sorting tab. In the Expression column enter =CountRows() and in the Direction column choose Descending.

Run application. This time the top 10 most popular documents are displayed in sorted order.

Step 7 – Add a chart

7.1
From the Toolbox drag a chart and drop it below the table. Right-click on the chart and change the chart type to ‘Simple Bar’.
7.2
From the Data Sources window drag the AccessTime field and drop it in the Category fields drop area.

7.3
Right-click on the chart and choose ‘Properties’. In the Chart Properties dialog switch to the Data tab. Press the Add button next to the Values list. In the Edit Chart Value dialog enter =CountRows() as the value, and then click OK to close the dialog.
7.4
In the Data tab of the Chart Properties dialog, press the Edit button next to Category groups list. In the Grouping and Sorting Properties dialog, notice that the expression =Fields!AccessTime.Value is present in the Expression list as well as the Label textbox. Modify both to =Fields!AccessTime.Value.Hour. Switch to the Sorting tab. Enter =Fields!AccessTime.Value.Hour under Expression. Choose Descending under Direction. Press OK to close the dialog.
7.5
Switch to X Axis tab and enter “Hour” as the title. Switch to Y Axis tab and enter “Hits” as the title. Switch to Legend tab and uncheck the ‘Show legend’ checkbox.

7.6
Run the application. The chart displays the number of hits by hour.

Step 8 – Style the report

8.1
Specify fonts and colors to make the report look attractive.
References

IIS Log File Formats
Appendix A – log parser source code

Imports System.IO

Imports System.Web

Imports System.Globalization

''' <summary>

''' Records the details of a single web server access.

''' </summary>

Public Class LogFileEntry

    Private m_document As String   ' The document that was accessed.

    Private m_accessTime As Date   ' The date and time of access.

    ''' <summary>

    ''' Constructs a new instance of LogFileEntry class.

    ''' </summary>

    ''' <param name="accessTime">The date and time of access.</param>

    ''' <param name="document">The document that was accessed.</param>

    Public Sub New(ByVal accessTime As Date, ByVal document As String)

        Me.m_accessTime = accessTime

        Me.m_document = document

    End Sub

    ''' <summary>

    ''' Date and time the web page was accessed.

    ''' </summary>

    ''' <value>Date and time of access.</value>

    Public ReadOnly Property AccessTime() As Date

        Get

            Return m_accessTime

        End Get

    End Property

    ''' <summary>

    ''' The document that was accessed.

    ''' </summary>

    ''' <value>The document accessed.</value>

    Public ReadOnly Property Document() As String

        Get

            Return m_document

        End Get

    End Property

End Class

''' <summary>

''' Parser for IIS log files.

''' </summary>

''' <remarks>See http://www.w3.org/TR/WD-logfile.html for log format.</remarks>

Public Class LogParser

    Private m_dateFieldIndex As Integer

    Private m_timeFieldIndex As Integer

    Private m_uriStemFieldIndex As Integer

    ' A list of filename extensions that should be ignored.

    Private m_ignoredExtensions As String() = {".gif", ".jpg", ".dll", ".js", ".css"}

    ''' <summary>

    ''' Parses the fields directive.

    ''' </summary>

    ''' <param name="line">The line that contains the fields directive.</param>

    ''' <remarks>Not all log files contain all fields, and fields may not be

    ''' in a pre-defined order. The log file contains a directive that lists

    ''' the fields present, and the order in which the fields appear. This 

    ''' method is used for extracting the field order from that line.</remarks>

    Private Sub ParseFieldsDirective(ByVal line As String)

        m_dateFieldIndex = -1

        m_timeFieldIndex = -1

        m_uriStemFieldIndex = -1

        Dim words As String() = line.Split(New Char() {" "}, StringSplitOptions.RemoveEmptyEntries)

        Dim i As Integer

        For i = 1 To words.Length - 1

            Select Case words(i)

                Case "date"

                    m_dateFieldIndex = i - 1

                Case "time"

                    m_timeFieldIndex = i - 1

                Case "cs-uri-stem"

                    m_uriStemFieldIndex = i - 1

            End Select

        Next

        If (m_dateFieldIndex = -1) Or (m_timeFieldIndex = -1) Or (m_uriStemFieldIndex = -1) Then

            Throw New ApplicationException("A required field is missing.")

        End If

    End Sub

    ''' <summary>

    ''' Parses a directive.

    ''' </summary>

    ''' <param name="directive">The directive string.</param>

    ''' <remarks></remarks>

    Private Sub ParseDirective(ByVal directive As String)

        If directive.StartsWith("#Fields:") Then

            ParseFieldsDirective(directive)

        End If

    End Sub

    ''' <summary>

    ''' Parses a log file entry.

    ''' </summary>

    ''' <param name="line">The log file line to parse.</param>

    ''' <returns>A LogFileEntry object if line was parsed successfully, or Nothing otherwise.</returns>

    ''' <remarks></remarks>

    Private Function ParseLogEntry(ByVal line As String) As LogFileEntry

        Dim fields As String() = line.Split(New Char() {" "}, StringSplitOptions.RemoveEmptyEntries)

        Dim document As String = HttpUtility.UrlDecode(fields(m_uriStemFieldIndex))

        ' If the file name ends with an extension that should be ignored then return nothing.

        Dim dotIndex = document.LastIndexOf(".")

        If dotIndex <> -1 Then

            Dim extension = document.Substring(dotIndex)

            If Array.IndexOf(m_ignoredExtensions, extension.ToLower()) <> -1 Then

                Return Nothing

            End If

        End If

        ' Concatenate the date and time fields in order to parse it.

        Dim dateTimeString As String = fields(m_dateFieldIndex) & " " & fields(m_timeFieldIndex)

        ' Parse the date and time using the format specified in the log format specification.

        Dim accessTimeUTC As Date = Date.ParseExact(dateTimeString, "yyyy-MM-dd HH:mm:ss", _

                  CultureInfo.InvariantCulture.DateTimeFormat)

        ' Convert the time from UTC to current time zone.

        Dim accessTime As Date = accessTimeUTC.ToLocalTime

        ' Return a new LogFileEntry object.

        Return New LogFileEntry(accessTime, document)

    End Function

    ''' <summary>

    ''' Parses a log file and returns its contents.

    ''' </summary>

    ''' <param name="logFilename">Name of log file.</param>

    ''' <returns>A collection of LogFileEntry objects.</returns>

    ''' <remarks></remarks>

    Public Function Parse(ByVal logFileName As String) As List(Of LogFileEntry)

        ' A collection representing all valid entries in the log file.

        Dim logFileEntries As New List(Of LogFileEntry)

        ' Read the contents of the file line by line and parse each line.

        Using reader As New StreamReader(logFileName)

            Do

                Dim line As String = reader.ReadLine()

                ' If there are no more lines then exit the loop.

                If (line Is Nothing) Then Exit Do

                ' If the first character is a # then this is a directive otherwise it is an entry.

                If line.StartsWith("#") Then

                    ParseDirective(line)

                Else

                    Dim access As LogFileEntry = ParseLogEntry(line)

                    ' If the line was successfully parsed add the LogFileEntry object to a collection.

                    If access IsNot Nothing Then

                        logFileEntries.Add(access)

                    End If

                End If

            Loop

        End Using

        ' Return the collection of LogFileEntry objects.

        Return logFileEntries

    End Function

    ''' <summary>

    ''' Parses supplied log file.

    ''' </summary>

    ''' <param name="logFile">Name of file to parse.</param>

    ''' <returns>Web accesses logged in the file.</returns>

    Public Shared Function ParseLogFile(ByVal logFile As String) As List(Of LogFileEntry)

        Dim parser As New LogParser

        Return parser.Parse(logFile)

    End Function

End Class

